


# Using formulae and functions in Microsoft Excel 2003

This guide provides an introduction to doing calculations in Excel 2003. It is assumed that the reader is familiar with the basics of Excel covered in *Guide 33: An introduction to Microsoft Excel 2003*.

A sample workbook referred to in the document is provided on the ITS Networked PC Service. Users of stand-alone PCs can get a copy from the ITS Web pages.



Document code: **Guide 35**  
Title: **Using formulae and functions in Microsoft Excel 2003**  
Version: **1.0**  
Date: **June 2006**  
Produced by: **University of Durham Information Technology Service**

**Copyright © 2006 University of Durham Information Technology Service**

### **Conventions:**

In this document, the following conventions are used:

- A typewriter font is used for what you see on the screen.
- A **bold typewriter font** is used to represent the actual characters you type at the keyboard.
- A *slanted typewriter font* is used for items such as filenames which you should replace with particular instances.
- A **bold font** is used to indicate named keys on the keyboard, for example, **Esc** and **Enter**, represent the keys marked Esc and Enter, respectively.
- A **bold font** is also used where a technical term or command name is used in the text.
- Where two keys are separated by a forward slash (as in **Ctrl/B**, for example), press and hold down the first key (**Ctrl**), tap the second (**B**), and then release the first key.

## Contents

<b>1</b>	<b>Creating formulae</b>	<b>1</b>
1.1	Entering a formula	1
1.2	Order of priority of operators	2
1.3	Editing formulae	2
1.4	Experimenting with parentheses	3
1.5	Constructing your own formula	3
1.6	Using cell references	3
1.6.1	Typing the cell references in a formula	3
1.6.2	Clicking on the cells to be included in a formula	4
1.7	Copying a formula	4
1.8	Absolute cell references	5
<b>2</b>	<b>Adding numbers</b>	<b>6</b>
2.1	Adding a column of numbers using the $\Sigma$ button	6
2.2	Adding a column of numbers using the SUM function	7
2.3	Adding rows and columns of numbers	7
<b>3</b>	<b>Functions</b>	<b>7</b>
3.1	Entering functions	8
<b>4</b>	<b>Mathematical functions</b>	<b>9</b>
4.1	SUM	9
4.2	ROMAN	10
4.3	Random numbers	11
<b>5</b>	<b>Text functions</b>	<b>11</b>
5.1	REPT	11
5.2	TRIM	12
5.3	CONCATENATE	12
5.4	UPPER, LOWER and PROPER	13
<b>6</b>	<b>Dates and times</b>	<b>13</b>
6.1	Date systems	14
6.2	Entering dates and times	14
6.3	How Excel interprets dates	14
6.4	Changing the default settings	15
6.4.1	To display four-digit years	15
6.4.2	To change the interpretation of two-digit years	15
6.5	Doing calculations with dates and times	15
6.5.1	TODAY	15
6.5.2	NOW	16
6.5.3	WEEKDAY	16
6.5.4	Adding and subtracting dates and times	17
6.6	Specialist date and financial functions	17
<b>7</b>	<b>Logical functions</b>	<b>17</b>
7.1	Conditional tests	18
7.2	IF	19
7.3	Nested IF statements	20
7.4	AND, OR and NOT	21
<b>8</b>	<b>Statistical functions</b>	<b>22</b>
8.1	AVERAGE	23
8.2	MAX	23
8.3	COUNT	24
8.4	COUNTBLANK	24
8.5	SUMIF	25

8.6	COUNTIF .....	25
8.7	SUMPRODUCT .....	25
<b>9</b>	<b>Lookup functions.....</b>	<b>26</b>
9.1	CHOOSE .....	26
9.2	VLOOKUP .....	27
9.3	TRANSPOSE.....	29
<b>10</b>	<b>Reference functions .....</b>	<b>30</b>
10.1	ADDRESS .....	30
10.2	OFFSET.....	31
<b>11</b>	<b>Functions as arguments to other functions.....</b>	<b>31</b>
<b>12</b>	<b>Errors .....</b>	<b>31</b>
12.1	Trapping errors using logical functions.....	32
12.2	Tracing errors .....	33
<b>13</b>	<b>Using labels instead of cell references.....</b>	<b>34</b>
<b>14</b>	<b>Working with names .....</b>	<b>36</b>
14.1	Names of cells and ranges .....	36
14.2	Defining a name using the Name box .....	36
14.3	Creating names from text cells .....	37
14.4	Pasting names into formulae .....	37
14.5	Applying names to existing formulae.....	38
14.6	Moving to a named cell or range .....	38
14.7	Producing a list of names .....	39
<b>15</b>	<b>Working with arrays .....</b>	<b>39</b>
15.1	Examples .....	39
15.2	Changing cells in an array range.....	40
15.3	Array constants.....	41
<b>16</b>	<b>Miscellaneous .....</b>	<b>41</b>
16.1	Recalculation .....	41
16.2	Formula bar formatting .....	42
16.3	Replacing formulae with their resulting values .....	42
16.4	Circular references .....	43
16.5	3-D formulae .....	43
16.6	Euro currency tools.....	43
16.7	R1C1 reference style.....	44
<b>17</b>	<b>Analysis Toolpack .....</b>	<b>45</b>
17.1	Analysis ToolPak not installed.....	45
17.2	Using the Analysis ToolPak.....	45
<b>18</b>	<b>What-if analysis .....</b>	<b>45</b>
<b>19</b>	<b>Other Excel documentation .....</b>	<b>46</b>

## 1 Creating formulae

The real strength of a worksheet lies in its ability to carry out calculations. A worksheet that does not contain calculations could have been created using a word-processing package instead.

An Excel workbook called **Data\_for\_Calculations.xls** has been prepared for you to use while working through this Guide.

If you are using a stand-alone PC, you may like to get a copy of this from the ITS WWW pages under **Information | Guides | Sample Files** (<http://www.dur.ac.uk/its/info/guides/files/excel/>).

If you are using the Networked PC service:


- 1 Activate **Excel**.
- 2 From the **F**ile menu, select **O**pen.
- 3 Click on the ▼ to the right of the **L**ook in: box.
- 4 Select the drive called **w**orkfile on 'dudley'(T:).
- 5 Double-click on the folder **i**ts.
- 6 Double-click on the folder **E**xcel.
- 7 Select the file **D**ata\_for\_Calculations.xls and click the **O**pen button.

### 1.1 Entering a formula

All formulae begin with an equals (=) sign. If you forget to type the =, the rest of the line will probably be entered into the cell as a piece of text.

- 1 In the **Data\_for\_Calculations** workbook, click in **B2** and type  
**=8+4**

The formula will be displayed on the formula bar.



	A	B	C	D
1				
2		=8+4		

- 2 Press the **Enter** key (or one of the arrow keys) to **lock** the formula in.

The cell **B2** now contains a formula (**=8+4**), but the answer **12** is displayed on-screen.

- 3 Click in **B2** to check the formula.

Formulae can contain values, cell references, mathematical operators and functions.

## 1.2 Order of priority of operators

The mathematical operators are shown below.

-	Negation (negative number)
%	Percent
^	Exponentiation (to the power of)
* and /	Multiplication and division
+ and -	Addition and subtraction
&	Text joining
= < > <= >= <>	Comparison

They are listed in decreasing order of priority starting with negation which has the highest priority (done first) and ending with comparison which has the lowest (done last). If a formula contains operators with the same priority, they are evaluated from left to right.

If you want to alter the order of evaluation, use parentheses (brackets) to group expressions. Any parts of a formula that are in parentheses are done first.

It is important to fully understand the order in which Excel carries out a calculation.

- 1 Click in **B3** and type  
 $=2*9+12/6-4$
- 2 Press the **Enter** key.
- 3 Make sure that you understand why the answer is **16**.

## 1.3 Editing formulae

A formula can be edited in the same way as any other entry in a cell.

To delete a cell reference or any other character, select it and press the **Backspace** or **Delete** key.

To replace a cell reference, select it and click in the replacement cell.

To add to the formula, position the insertion point where the extra characters are to go and type them in.

To undo changes that have *not* been locked in, press the **Esc** key or click the **Cancel** button (red cross) on the formula bar.

To undo changes that *have* been locked in, click the **Undo** button on the Standard toolbar, press **Ctrl/Z** or select **Undo** from the **Edit** menu.

## 1.4 Experimenting with parentheses

Now try editing the formula in **B3**. Put parentheses round different parts of the expression, as suggested below, and see how they affect the answer.

When you have more than one pair of brackets in a formula, Excel displays them in different colours on the formula bar.

Formula	Answer
$= (2 * 9) + 12 / (6 - 4)$	24
$= 2 * (9 + 12) / 6 - 4$	3
$= (2 * 9 + 12) / 6 - 4$	1
$= 2 * (9 + 12 / (6 - 4))$	30

## 1.5 Constructing your own formula

In order to practise creating a formula, try the following:

- 1 Think of an integer (whole) number. (For checking purposes it is probably advisable to make it less than 100.)
- 2 Enter a calculation in **E3** that will carry out *all* the following steps:
  - add **20** to the number you thought of
  - multiply the result by **4**
  - subtract **8** from the result
  - divide that result by **2**
  - subtract **twice** the number you thought of

The answer should be **36**. If it isn't, check your formula, paying particular attention to the brackets, and edit it as necessary.

## 1.6 Using cell references

In the examples so far, you have used actual values. Formulae can also contain cells references. You can practise using cell references by adding up five pairs of numbers.

The range **B10:C14** already contains the data shown below.

7	100
12	216
15	130
19	80
24	50

### 1.6.1 Typing the cell references in a formula

First, try *typing* the cell references in a formula:

- 1 Click in **D10** and type

**=B10+C10**

- 2 Press the **Enter** key.

(You could type **=b10+c10** if you prefer. Excel does not mind whether the cell reference is in upper or lower case letters.)

- 3 Check that the answer is **107**.

### 1.6.2 Clicking on the cells to be included in a formula

Now, try *clicking* on cells instead of typing their cell references:

- 1 Click in **D11** and type

**=**

- 2 Click in **B11** and note how Excel enters that reference into the formula.

- 3 Type

**+**

- 4 Click in **C11**.

- 5 Press the **Enter** key.

- 6 Check that the answer is **228**.

### 1.7 Copying a formula

A formula can be copied to another cell in the same way as text or numbers.

- 1 Click in **D11**.
- 2 Click the **Copy** button.
- 3 Click in **D12**.
- 4 Click on the **Paste** button (and ignore the smart tag if it appears).

Another way of copying is to use the **Auto Fill** handle:

- 1 Click in **D12**.
- 2 Move the cursor to the bottom right-hand corner of the cell. The cursor shape will change to a small black plus (+) sign.
- 3 Hold down the mouse button, drag to cell **D14** and then release the button.

The formula in D11 was **=B11+C11**. When that was copied and pasted into D12 it became **=B12+C12**. This is because Excel, unless instructed otherwise, works with *relative* cell references. It thinks about where cells are in relation to the cell in which the formula is entered.

This means that Excel thinks of that particular formula as

“take the contents of the cell two columns to the left of the cell containing the formula *then add* the contents of the cell one column to the left of the cell containing the formula”.

## 1.8 Absolute cell references

Sometimes you will need to work with absolute cell references that refer to cells by their fixed position in the worksheet.

- 1 Scroll down the worksheet until you can see the range **B20:F24** with data about VAT.

First, complete the **VAT** column:

- 1 Click in **C21**.
- 2 Type

**=B21\*F21**

and press the **Enter** key.

The formula in **C21** gives the correct answer (£19.60) but is not suitable for copying. To demonstrate this,

- 3 Copy the formula in **C21** to **C22**.

The value shown will be £0.00. This is because **F21** became **F22** when copied and the value in **F22** is zero.

When you are going to copy a formula and do not want a cell reference to change, you should use an absolute cell reference. A dollar symbol, **\$**, is used to fix a reference. When placed in front of a column name it fixes the column and in front of a row number it fixes the row.

So, for example, **\$F\$21** means *always* column **F** and *always* row **21**.

Note: If you click anywhere within a cell reference in a formula, and press the **F4** button, dollar symbols will be inserted for you automatically. You can have mixed references such as **\$Z34** (column is fixed, row is relative) and **Z\$34** (column is relative, row is fixed). Keep pressing **F4** if you wish to have mixed references.

Now change the formula in **C21**.

- 1 Double-click in **C21**.
- 2 Edit the formula to read

**=B21\*\$F\$21**

- 3 Press **Enter**.
- 4 Copy the formula in **C21** down to cell **C24** and check that the resulting values are correct.

Finally, complete the **Cost with VAT** column.

- 1 Click in **D21** and enter a formula for the cost including VAT (cost without VAT plus the VAT to be paid on each item).
- 2 Copy that formula down to **D24**.

Your results should be as shown below.

Cost without VAT	VAT	Cost with VAT
£112.00	£19.60	£131.60
£200.00	£35.00	£235.00
£119.00	£20.83	£139.83
£224.00	£39.20	£263.20

VAT rate
17.50%

When entering formulae you can, if you wish, refer to cells in other worksheets of the same workbook and to cells in other workbooks.

## 2 Adding numbers

There are a couple of easy ways of adding up a column of numbers. To access some suitable data,

- 1 Click on the worksheet tab **Sneezes**.

You will see a table with the heading **Sneeze count for Monday**. The totals are to be filled in as described below.

### 2.1 Adding a column of numbers using the $\Sigma$ button

Try using the **AutoSum** ( $\Sigma$ ) button to add up a column of numbers:

- 1 Highlight the cells from **C7** to **C13** inclusive. (Note that Excel displays the message Sum=128 on the status bar.)
- 2 Click on the **AutoSum** ( $\Sigma$ ) button on the toolbar.
- 3 Click away from the selected area.

The answer **128** is displayed in **C13**; the formula in that cell is actually **=SUM(C7:C12)**.

- 4 Click in **C13** to see that formula.

**SUM** is one of the built-in functions in Excel, probably the one most often used.

If you are sure that there is no ambiguity as to which cells are to be included in the calculation, you can simply select the cell below the numbers that are to be used and click the **AutoSum** button.

**Note:** Clicking on the arrow next to the **AutoSum** button gives quick access to the functions **Average**, **Count**, **Max** and **Min**, and to all the built-in functions (select **More Functions**).

## 2.2 Adding a column of numbers using the SUM function

Now try using the **SUM** function:

1 Click in **D13**.

2 Type

**=SUM(D7:D11)**

3 Press the **Enter** key.

If you wish, instead of typing **D7:D11** you could just highlight those particular cells. Excel would fill in the description of the range and supply the closing bracket.

## 2.3 Adding rows and columns of numbers

1 Scroll down the worksheet to the table with the heading **Sneeze count for Tuesday**.

The row **Totals** and the column **Number of sneezes in eight hours** can be filled in quickly by using the **AutoSum** button. Again, you select your data plus the area where the totals are to appear as follows:

2 Highlight the cells from **C20** to **E26** inclusive.

3 Click on the **AutoSum** ( $\Sigma$ ) button on the toolbar.

4 Click away from the selected area.

This procedure is often used when adding up tables of costs (currency).

## 3 Functions

A function is a predefined formula that operates on one or more values and returns one or more values.

You have seen how using the **SUM** function can save you a lot of typing.

For example, the formula

**=SUM(A1:A15)**

is considerably shorter than

**=A1+A2+A3+A4+A5+A6+A7+A8+A9+A10+A11+A12+A13+A14+A15**

All functions consist of a function name (such as **SUM**) followed by a set of arguments (separated by commas) in brackets. The arguments specify the values or cells to be used by the function.

The maximum number of arguments allowed in a function is 30. In the example above, there is just one argument, namely, **A1:A15**.

A few functions have no arguments but the brackets are still needed, as in the following examples:

**=2\*PI()**

which calculates the value of  $2\pi$

**=RAND()**

which returns a random number that is greater than or equal to 0, and less than 1.

If you enter a space between the name of a function and the opening bracket, Excel will warn you that it has found an error.

Excel has more than 300 built-in functions. Some of them perform complex calculations but the examples that follow are of the more straightforward kind.

If there is not a built-in function that does precisely what you need, you can actually create your own custom functions but that involves using Visual Basic code and will not be discussed in this document.

### 3.1 Entering functions

You can enter a function into a worksheet by

- typing it in (as you did in section 2.2)
- clicking the **Insert Function (fx)** button (to the left of the formula bar)
- selecting **Insert | Function**

If you type in the name of the function, use lower case letters. Then, when you either press the Enter key or select another cell, Excel will change the name to uppercase providing it has been typed correctly. So, if lowercase letters remain, you will be alerted to the fact that you have typed the name incorrectly and you can correct it.

When you click the **Insert Function** button or select **Insert | Function**, the **Insert Function** dialog box will be displayed.


If you know the name of the function you want to use but cannot remember which category it is in, in the **Insert Function** dialog box, *either*,

type the name in the **Search for a function:** box and click **Go**

*or*,

select **All** in the **Or select a category:** box (this gives a complete alphabetical list of available functions).

If you feel that you might need extra help in creating your function, click on the **Help on this function** button in the bottom left corner of the **Insert Function** dialog box. The **Microsoft Excel Help** window will then be displayed (as in the picture that follows).



To illustrate how to build up a function, examples of various types of function will now be considered.

## 4 Mathematical functions

### 4.1 SUM

First add up the numbers in the **Number of sneezes in eight hours** column of the **Sneeze count for Monday** table as follows:

- 1 Make sure that the **Sneezes** worksheet is still the active one.
- 2 Scroll up to the first table, **Sneeze count for Monday**.
- 3 Click in **E13**.
- 4 Click the **Insert Function (fx)** button, to the left of the formula bar.

The **Insert Function** dialog box will open.


- 5 In the **Or select a category:** box, select **Math & Trig**.
- 6 In the **Select a function:** box, scroll until you can select **SUM**.

Note that a description of the selected function is given in the **Insert Function** dialog box.

- 7 Click on **OK**.

The **Function Arguments** dialog box appears. This contains one edit box for each argument of your function.

Note how Excel has tried (but failed) to guess what you want to do and has filled in the range **C13:D13** (which would actually give the same result).



- 8 If necessary, move the **Function Arguments** dialog box (just drag it) so that you can see cells **E7:E11**.
- 9 Select cells **E7:E11** and note how Excel fills in that information in the **Number1** box.

Sometimes you may prefer to type in the range description yourself rather than use the selection technique.

- 10 Click on the **OK** button.
- 11 Check that the resulting total is correct (**199**).

**Note:** Occasionally, you may need to see more of your worksheet than can be achieved by just moving the Function Arguments dialog box. Whenever you wish, a dialog box can be shrunk as follows:



- Click on the **collapse dialog** button at the right-hand end of the edit box. (The dialog box will collapse.)
- Select your cells.
- Click the **collapse dialog** button again. (The dialog box will return to its original size.)

## 4.2 ROMAN

Suppose you want to convert some arabic numbers to roman (as text) but you do not know whether there is a function that will do this.

- 1 Click in **D32**.
- 2 Click the **Insert Function (fx)** button.
- 3 In the **Search for a function:** box, type **roman** and click **Go**.

You will see the function name **ROMAN** listed in the **Select a function:** pane, and, below that, a description of what that particular function does.

- 4 Make sure that **ROMAN** is selected and click on **OK**.
- 5 In the **Function Arguments** dialog box, type **C32** in the **Number** pane.

- 6 Click on **OK**.
- 7 Copy that formula down through cells **D33:D38**.

Arabic number	Roman number
5	V
6	VI
21	XXI
40	XL
50	L
100	C
900	CM

Throughout the rest of this document instructions will be given for entering a function by typing it in, rather than by using the **Insert Function** dialog box. You can use whichever method suits you best.

### 4.3 Random numbers

Excel provides functions for generating random numbers.

**RAND()** returns a random number, greater than or equal to 0, and less than 1

**RANDBETWEEN(*bottom*,*top*)** returns a random integer number, greater than or equal to *bottom*, and less than or equal to *top*

For both functions, the number generated changes on recalculation.

If you want to generate a set of random numbers and then freeze them:

- use RAND or RANDBETWEEN, select the formulae, copy them, select Edit|Paste Special and choose Values (see section 16.3)
- use the Tools|Data Analysis|Random Number Generation tool (see section 17) which produces constants

## 5 Text functions

So far, you have only dealt with numbers. There are several functions that enable you to manipulate text strings, convert numeric entries into text strings and convert numeric text entries into numbers.

To access some suitable data,

- 1 Click on the **Text** tab of the **Data\_for\_Calculations** workbook.

### 5.1 REPT

The **REPT** function gives you a quick way of filling a cell with a string of characters repeated a specified number of times.

**General form:**

**=REPT(*text*,*number\_of\_times*)**

**Example 1:** Create a text string containing 107 asterisks

1 Click in **B3**.

2 Type

**=rept("\*\*\*",107)**

and press **Enter**.

A row of asterisks should appear on the worksheet.

**Example 2:** Repeat some text eight times

1 Click in **B17**.

2 Type

**=rept("please cut here.... ",8)**

and press **Enter**.

## 5.2 TRIM

Leading or trailing spaces often prevent entries in a worksheet from sorting correctly. The **TRIM** function eliminates leading, trailing, and extra blank characters from a string, leaving only a single space between words.

**General form:**

**=TRIM(*text*)**

**Example:** Create a tidy version of some untidy text

1 Click in **B2** and look at the formula bar to see the extra spaces at the beginning of the heading and between the words.

Now create a neater version on row 4.

2 Click in **B4**.

3 Type

**=trim(b2)**

and press **Enter**.

**Note:** The **CLEAN** function is similar to **TRIM** except that it deals with non-printable characters like tabs and program-specific codes. This can be particularly useful when dealing with data imported from other programs.

## 5.3 CONCATENATE

This function can be used to build up a long string from shorter strings. It has the same effect as using the **&** operator mentioned in section 1.2.

**General form:**

**=CONCATENATE(*text1,text2,...*)**

with up to 30 text arguments.

**Example:** Join three text strings

Create each person's name from their first name and their surname.

1 Click in **D7**.

2 Type

**=concatenate(c7," ",b7)**

and press **Enter**.

The space between the quote marks is needed to give a space between the two parts of the resulting name. Without it you would get **peterlee**.

3 Copy the formula in **D7** down through cells **D8:D15**.

#### 5.4 UPPER, LOWER and PROPER

These three functions alter the case of characters in text strings. UPPER converts text to uppercase and LOWER converts text to lowercase. PROPER capitalises the first letter of each word (and any other letters in the string that do not follow another letter) and all other letters are converted to lowercase.

**General forms:**

**=UPPER(text)**

**=LOWER(text)**

**=PROPER(text)**

**Example 1:** Improve the appearance of the names in **D7:D15**.

1 Click in **E7**.

2 Type

**=proper(d7)**

and press **Enter**.

3 Copy that formula down through cells **E8:E15**.

4 Check the capitalisation is correct.

**Example 2:** Capitalise the birthday greeting.

1 Click in **C20**.

2 Type

**=upper(b20)**

and press **Enter**.

## 6 Dates and times

In Excel, the basic unit of time is the day. Dates are stored as sequential numbers known as **serial values**.

## 6.1 Date systems

Excel supports two date systems, the **1900** and the **1904** date systems. The default system for Microsoft Excel for Windows is 1900 so the date serial value 1 corresponds to January 1<sup>st</sup>, 1900. The maximum value 2958465 represents December 31<sup>st</sup>, 9999.

Excel for a Macintosh uses the 1904 system and its date serial value 1 corresponds to January 2<sup>nd</sup>, 1904. If a file created in Excel version 2 or later on a Mac is opened in Excel for Windows, dates are automatically changed to the 1900 date system. Similarly, when an Excel for Windows file is opened on a Mac, the date system is changed to 1904. However, care should be taken when transferring workbooks between a PC and a Mac.

If necessary, the date system used in calculations can be changed. From the **Tools** menu, select **Options**; click the **Calculation** tab and under **Workbook options** select the date system you require.

## 6.2 Entering dates and times

The time of day is a decimal value that represents the fraction of a day between when it began (midnight) and the specified time. So, noon is represented by 0.5 (half a day).

The way that a date or time is displayed in a cell depends on which number format has been applied to the cell. The number format chosen does not affect the actual cell value.

If Excel recognises that you have typed a date or time, it changes the formatting of the cell from **General** number to one of the built-in date/time formats (these are right-aligned).

If Excel does not recognise your date or time, it will be entered as text (left-aligned in the cell).

When you want to enter a date and a time in the same cell, just separate them with a space. If your time is based on the 12-hour clock, type a space followed by **AM** (or **A**), **PM** (or **P**). Otherwise Excel assumes you are using a 24-hour clock.

The serial value of a date and the decimal fraction of a time can be displayed in a cell by changing that cell's format to the **General** format.

## 6.3 How Excel interprets dates

When two digits are entered for the year, Excel interprets them as follows:

<b>00 to 29</b>	<b>2000 to 2029</b>
<b>30 to 99</b>	<b>1930 to 1999</b>

This default can be changed (see section 6.4.2).

To be really sure that Excel is interpreting a year correctly, type all four digits (for example, 2019 rather than 19). You should be as precise as possible when entering dates and try to avoid ambiguous situations.

For example,

**14-Dec** is 14<sup>th</sup> December this year

**December 14** is 1<sup>st</sup> December in 2014

**32-Dec** is not accepted as a date (a green flag appears in the corner of the cell with an associated error type smart tag)

## 6.4 Changing the default settings

### 6.4.1 To display four-digit years

When you enter a date, it is automatically formatted to display a two-digit year. On a stand-alone PC you can change the default date format so that it shows four-digit years instead.

From the **Start** menu, point to **Settings** and then click **Control Panel**. Double-click **Regional Settings** (or Regional Options) and click the **Date** tab. In the **Short date format** list, click on one of the formats that uses four digits for the year ("yyyy").

### 6.4.2 To change the interpretation of two-digit years

If you are using Windows 98 or later on a stand-alone PC you can change the century that Excel (and *all* your Windows programs) assumes for a two-digit year.

From the **Start** menu, point to **Settings** and click **Control Panel**. Double-click **Regional Settings** (or Regional Options) and click the **Date** tab. In the **When a two-digit year is entered, interpret as a year between** box, change the setting for the upper limit for the century (the lower limit changes automatically).

Note: The characters recognised as date and time separators are also set in the Regional Settings.

## 6.5 Doing calculations with dates and times

Since dates and times are values, they can be added, subtracted and included in calculations. When entering a date or time in a calculation, put it inside quotation marks.

For example,

**= "21/8/2000" - "18/7/2000"**

gives the answer **34**.

You may like to experiment now with some dates and times.

### 6.5.1 TODAY

- 1 Click on the **Date&Time** tab of the workbook.
- 2 Click in **C2**.

3 Type

**=today()**

and press **Enter**.

Today's date will be displayed in the **d/m/yy** format where **d** represents the day, **m** the month and **yy** the last two digits of the year.

When typing in a date, it is all right to use just the last two digits of the year if your date is between 1930 and 2029. If however you are entering a date outside this range, you must type all four digits.

**Tip:** Press **Ctrl/;** to enter the current date in a cell or formula. To enter the current time, hold down the **Ctrl** and **Shift** keys and press the colon (:) key.

### 6.5.2 NOW

Next try using a combined date and time format:

1 Click in **C3**.

2 Type

**=now()**

and press **Enter**.

Today's date and the current time will be displayed in the **d/m/yy h:mm** format where **h** represents hour, and **mm** the minutes (date as before).

### 6.5.3 WEEKDAY

You may already know on which day of the week you were born. You can check that using the function WEEKDAY.

1 Enter your date of birth in **B7**.

2 Click in **B10**.

3 Click the **Insert Function** button.

4 In the **Or select a category:** box, select **Date & Time**.

5 In the **Select a function:** box, select **WEEKDAY**.

6 Click **OK**.

7 In the **Serial\_number** box of the **Function Arguments** dialog box, type **B7** (or click in that cell).

8 Click in the **Return\_type** box and read the information about Return\_type values.

9 Type **1** and click **OK**.

The answer will be a number in the range 1 to 7 (1=Sunday, 2=Monday and so on). Later, in section 9.1, you will learn how to show the actual name of the day.

#### 6.5.4 Adding and subtracting dates and times

Dates and times can be added and subtracted just like numbers. Have a look at the Railway timetable that starts in cell **B13**.

To calculate the duration of each journey:

1 Click in **D16**.

2 Type

**=C16-B16**

and press **Enter**.

3 Copy that formula down through cells **D17** to **D25**.

To see what the shortest (minimum) journey time is, use the statistical function **MIN**.

1 Click in **D27**.

2 Type

**=min(D16:D25)**

and press **Enter**.

That will show the shortest journey time as 00:15.

To show the time in minutes:

3 Click in **D28**.

4 Type

**=minute(D27)**

and press **Enter**.

The shortest journey time will be displayed as 15.

#### 6.6 Specialist date and financial functions

If you have installed the **Analysis ToolPak** add-in (see section 16.6), you will have available to you:

- a set of specialised date functions to help with things like payrolls and work schedules
- a set of financial functions to assist with business calculations (in addition to those available in the **Financial** function category)

### 7 Logical functions

A logical value is either **TRUE** or **FALSE**.

There are six logical operators

=	Equal to
---	----------

>	Greater than
<	Less than
>=	Greater than or equal to
<=	Less than or equal to
<>	Not equal to

## 7.1 Conditional tests

A conditional test is an equation that compares two items. These items can be numbers, functions, formulae, labels or logical values.

A conditional test must include at least one logical operator.

The result of a conditional test is either TRUE or FALSE.

A few examples would probably help to explain about conditional tests. To access some suitable data,

- 1 Click on the **Logical** tab of the workbook.

**Example 1:** Is the value in **B2** less than 4?

- 1 Click in **D7**.

- 2 Type

**=B2<4**

and press **Enter**.

The result should be **TRUE**.

**Example 2:** Is the sum of the values in **C2** and **C3** greater than or equal to 8?

- 1 Click in **D8**.

- 2 Type

**=C2+C3>=8**

and press **Enter**.

The result should be **FALSE**.

**Example 3:** Does the cell **B4** contain the text **Hello**?

- 1 Click in **D9**.

- 2 Type

**=B4="Hello"**

or

**=b4="hello"**

and press **Enter**. (It is not case sensitive.)

The result should be **TRUE**.

**Example 4:** Is the value in **C3** the same as the value in **C4**?

1 Click in **D10**.

2 Type

**=C3=C4**

and press **Enter**.

The result should be **FALSE**.

## 7.2 IF

The **IF** function is very useful. It will carry out a conditional test and then return one value if the condition is true but a different value if the condition is false.

**General form:**

**=IF(condition, value\_if\_true, value\_if\_false)**

**Example 1:** Quality of a mark

1 Look at the area of cells **F2:G5** on the **Logical** worksheet.

A mark is felt to be **Good** if it is greater than **49** and **Poor** otherwise.

2 Click in **G3** and enter

**=IF(F3>49,"Good","Poor")**

3 Copy that formula down through cells **G4** and **G5**.

The results should be **Good, Poor, Poor**.

**Example 2:** Coding

1 Look at the area of cells **I2:J6**.

The letters are to be coded — **1** for **M**, and **0** for all other letters.

2 Click in **J4** and enter

**=IF(I4="M",1,0)**

This will return the value **1** if **I4** contains **M**, otherwise it will return the value **0**.

3 Copy that formula down through cells **J5** and **J6**.

The results should be **0, 1, and 0**.

### Example 3: Exam results

- 1 Look at the table of examination results in **B13:K24**.

Suppose that a student is deemed to have passed if he/she gets a total mark of 200 or more but otherwise failed. Fill in the **Result 1** column accordingly:

- 2 Click in **I16**.

- 3 Type

**=if(H16>=200,"Pass","Fail")**

and press **Enter**.

- 4 Copy that formula down through cells **I17:I24**.

The results should be **Pass, Pass, Fail, Pass, Fail, Pass, Pass, Pass, Fail**.

### 7.3 Nested IF statements

Sometimes your problem may be so involved that you cannot solve it by using just the techniques you have met so far. An extremely useful feature is the ability to nest IF functions (put one inside another).

You can nest up to seven IF functions as long as the number of characters in your formula does not exceed the maximum number of characters that can be entered in a cell.

There is not a single general form but the following is typical:

**=IF(condition1,value1,IF(condition2,value2,value3))**

The three arguments for the first IF are:

- *condition1*
- *value1*
- *IF(condition2,value2,value3)*

The three arguments for the second IF are:

- *condition2*
- *value2*
- *value3*

The value taken by the cell containing this formula will be as shown in the table below.

<b>condition1</b>	<b>condition2</b>	<b>value</b>
TRUE		value1
FALSE	TRUE	value2
FALSE	FALSE	value3

As an example, look at the table of examination results in **B13:K24**.

The **Action** column is to be completed using the following rules:

- If a student has passed, the result is “No action”.
- If a student has failed (in the Result 1 column) but has a total greater than 99, he/she has to “Resit”.
- If a student has failed (in the Result 1 column) and has a total less than 100, he/she has to “Leave”.

Use a nested **IF** statement as follows:

1 Click in **K16**.

2 Type

**=if(I16="Pass","No action",if(H16>99,"Resit","Leave"))**

and press **Enter**.

3 Copy that formula down through cells **K17:K24**.

4 Check your results make sense.

#### 7.4 **AND, OR and NOT**

When building up a conditional test you sometimes need quite complicated conditions. These three functions can help to simplify the job. They use the logical operators listed at the beginning of section 7. The functions **AND** and **OR** can take up to 30 logical arguments whereas **NOT** has only one argument.

**General forms:**

**=AND(logical1,logical2,...,logical30)**

**=OR(logical1,logical2,...,logical30)**

**=NOT(logical1)**

The arguments for the functions can be

- conditional tests
- references to cells containing logical values
- arrays of logical values

##### **AND**

If *all* the arguments in an **AND** function are true, the result is **TRUE**.

If *one or more* arguments are false, the result is **FALSE**.

##### **OR**

If *at least one* of the arguments in an **OR** function is true, the result is **TRUE**.

If *all* the arguments in an **OR** function are false, the result is **FALSE**.

##### **NOT**

This function negates a condition.

If the argument in a **NOT** function is false, the result is **TRUE**.

If the argument in a **NOT** function is true, the result is **FALSE**.

A few very simple examples follow. Just read them and make sure you understand the result.

Formula	Result
=AND(5>3,80<96)	TRUE
=AND(2*24<67,7<0.34,4*4=2*8)	FALSE
=OR(4*4=8*2,56>297)	TRUE
=OR(2=3,3=4,4=5)	FALSE
=NOT(5<3)	TRUE
=NOT(6=6)	FALSE

In your own worksheets, you will probably use logical conditions involving some cell references rather than just values as above.

Have a look again at the table of examination results. Suppose that this time a student is said to have passed if he/she gets at least 50 marks in each module. (This means that good marks in one module cannot compensate for poor marks in another.)

This definition of pass/fail is to be entered in the **Result 2** column.

1 Click in **J16**.

2 Type

=if(and(D16>=50,E16>=50,F16>=50,G16>=50),"Pass","Fail")

and press **Enter**.

3 Copy that formula down through cells **J17:J24**.

The results should be **Fail, Pass, Fail, Pass, Fail, Pass, Fail, Pass, and Fail**.

## 8 Statistical functions

Excel provides a great deal of assistance when it comes to analysing statistical data. There are several built-in functions, a few of which will be looked at in this document. In addition, the Analysis ToolPak can be used (see section 16.6). This is an add-in module with a collection of functions and tools that enable you to produce rank-and-percentile tables, perform regression analysis, apply Fourier transformations to your data, and so on. Consult Excel's online Help for details.

Some data is available for you to experiment with statistical functions.

1 Click on the **Stats** worksheet tab.

## 8.1 AVERAGE

The AVERAGE function calculates the average, or mean, of the numbers in a range by adding the numeric values and then dividing by the number of values.

**General form:**

**=AVERAGE(range)** or  
**=AVERAGE(number1,number2,number3,...)**

Cells containing text or logical values, and empty cells, are ignored but cells containing a zero value are included.

**Example:** Calculate the average number of Caraway Thyme plants sold in a day.

1 Click in **H36** (the first green cell).

2 Type

**=average(D29:J29)**

and press **Enter**.

The answer will be 14.

This illustrates what the **AVERAGE** function does when one or more cells in the range do not contain data (**F29** in this example). The function adds up all the numbers (84) but divides by the number of cells that contain data (6 in this case) which gives 14.

If you wanted to add up all the numbers (84) but divide by the number of days in the week (7), you could use **=sum(D29:J29)/7** which would give the answer 12. Try entering that formula in **H37**.

**Note:** Excel provides an AVERAGEA function, which is the same as AVERAGE except that text and logical values *are* included in the calculation. See Excel's Help for information about other A functions such as COUNTA, MAXA, MINA, STDEVA, STDEVPA, VARA and VARPA.

## 8.2 MAX

This function returns the largest value in a given range.

**General form:**

**=MAX(range)** or  
**=MAX(number1,number2,number3,...)**

**Example:** Calculate the maximum number of Common Mint plants sold in a day.

1 Click in **H38**.

2 Type

**=max(D19:J19)**

and press **Enter**.

The answer should be 31. There is a corresponding function **MIN** that calculates the smallest value.

### 8.3 COUNT

This function works out how many cells in a given range contain numbers (including dates and formulae with numerical answers). It ignores blank cells and cells containing text, logical or error values.

**General form:**

**=COUNT(*range*)** or  
**=COUNT(*number1,number2,number3,...*)**

**Example:** Calculate how many different kinds of herb were sold on Thursday.

1 Click in **H39**.

2 Type

**=count(G13:G33)**

and press **Enter**.

The answer should be 9.

If you want to count *all* non-blank cells, irrespective of what they contain, use **COUNTA**.

### 8.4 COUNTBLANK

COUNTBLANK counts the number of empty cells in a range.

**General form:**

**=COUNTBLANK(*range*)**

You have to be careful with this function because a cell may sometimes appear to be blank when it isn't really. This could be because it contains the null string "", a space " ", or a formula with answer zero.

**Example:** Calculate the number of days when no Pink Hyssop was sold.

1 Click in **H40** and enter

**=countblank(D16:J16)**

The answer should be 3.

## 8.5 SUMIF

This function tests the value in each cell of a range before adding the contents of specified cells to a total.

**General form:**

**=SUMIF(*range\_to\_test*,*criteria*,*range\_to\_sum*)**

where

- *range\_to\_test* is the range to be tested
- *criteria* stipulates what the test is
- *range\_to\_sum* is the range containing the values to be summed if the test is satisfied

**Example:** Calculate the number of Thyme plants sold on Sunday.

Here we want to look down the first column of the table to find rows where the Name entered is **Thyme**, and then add the corresponding value in the **Sun** column to the total.

- 1 Click in **H41** and enter

**=sumif(B13:B33,"Thyme",J13:J33)**      or

**=sumif(B13:B33,"Thyme",Sun)**

The answer should be 112.

## 8.6 COUNTIF

This function counts the number of cells in a range that match specified criteria.

**General form:**

**=COUNTIF(*range*,*criteria*)**

**Example:**

Suppose you want to count how many cells in the **Sales of herbs** table contain values greater than 20.

- 1 Click in **H42** and enter

**=countif(D13:J33,">20")**

The answer should be 10.

## 8.7 SUMPRODUCT

When given two or more sets of numbers, this function multiplies the corresponding values together, adds up those products, and returns the sum. Any non-numeric values are treated as though they are zeros.

**General form:**

**=SUMPRODUCT(array1,array2,array3,...)**

**Example:**

Look at the table in the cells **B46:H53** giving dimensions of five panels that are to be painted. The total area to be painted can be obtained by adding up the areas (height multiplied by width) of the individual panels.

- 1 Click in **F52** and enter

**=sumproduct(D49:H49,D50:H50)**

The answer should be 48.06 (square metres).

Some related functions are:

**=SUMSQ(number1,number2,number3,...)** which adds up the squares of the numbers in the arguments

**=SUMX2PY2(x\_array,y\_array)** which calculates the sum of the sums of the squares of the corresponding values in the x and y arrays, which is  $\Sigma(x^2+y^2)$ .

**=SUMX2MY2(x\_array,y\_array)** which calculates  $\Sigma(x^2-y^2)$ .

**=SUMXMY2(x\_array,y\_array)** which calculates  $\Sigma(x-y)^2$ .

There are many other functions to help with statistical calculations. Browse through what is offered in the **Statistical** category of the **Insert Function** dialog box.

The Analysis Toolpak (see section 16.6) will enable you to do many things of a statistical nature such as

- produce rank-and-percentile tables
- extract random or periodic samples from a data set
- perform regression analysis
- apply Fourier transforms to data

## 9 Lookup functions

If you have information stored in a list or table, there are functions to help you.

To consider specific examples,

- 1 Click on the **LookUp** tab of the workbook.

### 9.1 CHOOSE

This function chooses a value or action to perform from a list of values, based on an index number.

### General form:

**=CHOOSE(index\_num,value1,value2,... )**

where the *values* can be up to 29 numbers, cell references, defined names, formulae, functions or text arguments.

### Example 1: Show season of year

- 1 Look at **Table 1** on the **LookUp** worksheet.

This shows some numbers corresponding to times of year (1=Spring; 2=Summer, and so on). The CHOOSE function can be used to produce the appropriate text, with the help of the words in G5:G8.

- 2 Click in **C6** and enter

**=choose(B6,\$G\$5,\$G\$6,\$G\$7,\$G\$8)**

The dollars (\$) are needed because the formula is going to be copied.

- 3 Click in **C6** and drag the formula down through cells **C7:C12** (use the Fill Handle).

The results should be **Autumn, Winter, Summer, Spring, Summer, Autumn, and Winter**.

### Example 2: Day of the week

- 1 Look at **Table 2**.
- 2 Click in **C15** and enter today's date (type it in or use the TODAY function — see section 6.5.1).
- 3 Click in **F15** and enter

**=weekday(C15,1)**

The result will be a number (1=Sunday, 2=Monday, and so on).

It would be better to display the name of the day (Sun, Mon, and so on). So,

- 4 Edit the formula in **F15** to read

**=choose(WEEKDAY(C15,1),"Sun","Mon","Tue","Wed","Thu","Fri","Sat")**

The correct day of the week should then be displayed.

## 9.2 VLOOKUP

The VLOOKUP function will retrieve information from a table (which can itself be part of a larger table). You do not have to know precisely where the value you want is stored. The function decides which row of the table is to be used — it finds the position, in the first column of the table, of the largest value that is less than or equal to a lookup\_value supplied by you. You tell it which column is to be used for retrieving the information you require.

### General form:

**=VLOOKUP(*lookup\_value*,*table\_array*,*col\_index*,*range\_lookup*)**

where

- *lookup\_value* is the value to be found in the first column of the table (it can be a value, reference or text string).
- *table\_array* is the array (or range name) that defines the table. If *range\_lookup* is TRUE, the values in the leftmost column of the table must be sorted in ascending order. If *range\_lookup* is FALSE, the leftmost column of the table does not have to be sorted.
- *col\_index* tells the function which column of the table to look in to find the function's result. This number must be greater than or equal to 1 and never greater than the number of columns in the table.
- *range\_lookup* is a logical value that determines whether *lookup\_value* has to be matched exactly (set *range\_lookup* to FALSE) or approximately (set *range\_lookup* to TRUE or omit it completely).

The function works with vertical tables — those where the comparison values are in the leftmost column. These values can be either numbers or text.

If your table is a horizontal one (with the comparison values in the first row), use the similar function called **HLOOKUP**.

### Example 1:

Look at **Table 3** on the **LookUp** worksheet and find the value in **Column 4** of the table corresponding to the largest value in **Column 1** that is not greater than 30.

Here **Column 1** refers to the first column in the table (not to the first column of the worksheet). Other columns in the table are counted relative to that first column.

Because you are looking for the largest value that is not greater than 30, the *lookup\_value* is 30.

- 1 Click in **E28** and enter the formula

**=vlookup(30,C20:F24,4)**

You should get the answer 18 as explained below.

The range **C20:F24** determines the extent of the table. In Column 1, the largest value that is not greater than 30 is 25 and that can be found in Row **3**. The **4** in the function indicates that Column 4 is to be used. The value in the table in row 3 and column 4 is the answer (18).

### Example 2:

Now experiment with **Table 4** (scroll down until you can see it).

- 1 Click in **C42** and type the name of one of the listed trees,

### Oak

- 2 Click in **C43** and enter a formula that will look up the tree in the table and return the answer as to whether it is Evergreen or Deciduous.  
Enter

**=vlookup(C42,B34:C39,2,FALSE)**

FALSE was used to get an exact match.

The result should be **Deciduous**.

- 3 Change the value in **C42** to a different tree, **Holly**, and check that the answer has automatically changed to **Evergreen**.

### Example 3:

If you would like to try using HLOOKUP (the corresponding function for a horizontal table), scroll down to **Table 5**.

- 1 Click in **C56** and type the name of one of the countries listed in the table.
- 2 Click in **C57** and enter a formula that will give the Gross Profit for the country in **C56**.
- 3 Check the answer is correct.
- 4 Change the value in **C56** and check that the value in **C57** changes automatically and correctly.

**Note:** If your comparison values and results are in separate areas of your worksheet, consider using the function **LOOKUP**.

## 9.3 TRANSPOSE

You can change the orientation of a rectangular array of values from vertical to horizontal (or vice versa). The first row of a horizontal array becomes the first column of the new vertical array (and so on).

### General form:

**=TRANSPOSE(array)**

This function must be entered as an **array formula** into a range that has as many rows and columns as the original array has columns and rows.

### Example:

Scroll down to **Table 6** in cells **B63:E75**. It shows distributions of word lengths in some letters signed Quintus Curtius Snodgrass that may have been written by Mark Twain. The table has 13 rows and 4 columns. It could be transposed and stored in **B79:N82** (4 rows by 13 columns) as follows:

- 1 Select the cells **B79:N82** (the location for the transposed table).
- 2 Type the following formula, but do *not* press the Enter key

**=transpose(B63:E75)**

- 3 Press the **Ctrl, Shift and Enter** keys together to enter the function in each cell of the selected range.
- 4 Check that you have achieved the desired result.

Another way of transposing a rectangular area of cells is to:

- Copy the cells.
- Click in the cell that is to be in the top left-hand corner of the transposed range.
- Select **Edit | Paste Special**.
- Tick the **Transpose** box and click **OK**.

## 10 Reference functions

If you wish to manipulate references, there are functions to help with that too. This document will not consider such functions in any detail but a couple of examples will give you an idea of the kind of thing that can be done.

### 10.1 ADDRESS

This function enables you to build a reference from numbers.

General form:

**=ADDRESS(row\_num,col\_num,abs\_num,a1,sheet\_name)**

where

- *row\_num* and *col\_num* determine the row and column values
- *abs\_num* determines whether the resulting address uses absolute references; values are

1	absolute
2	absolute row, relative column
3	relative row, absolute column
4	relative

- *a1* is a logical value;
  - TRUE** gives a resulting address in **A1** format
  - FALSE** gives a resulting address in **R1C1** format
- *sheet\_name* enables you to specify a sheet name (put double quotes around the text if it consists of more than one word).

For example:

**=ADDRESS(1,1,1,TRUE,"My data")** returns the reference **'My data'!\$A\$1**

## 10.2 OFFSET

This returns a reference of a specified height and width located at a position that is specified relative to some base reference.

General form:

**=OFFSET(*ref,rows,cols,height,width*)**

where

- *ref* is the base reference from which the offset is calculated
- *rows* specifies the vertical distance from the base reference to the new reference (positive value means below the base reference; negative above)
- *cols* specifies the corresponding horizontal distance (positive to the right, negative to the left of the base reference)
- *height* and *width* specify the shape of the resulting reference; they are optional but must be positive if included. When they are omitted, the resulting reference has the same dimensions as the base reference.

So, **=OFFSET(B5:E10,-3,2)** will return the reference **D2:G7**.

## 11 Functions as arguments to other functions

Functions, or indeed calculations involving several functions, can be used as arguments in another function.

As an example of this, consider the following.

The formula **=SQRT(*number*)** will calculate the square root of a number.

**=SQRT(9)** will return **3**

The formula **=POWER(*number,power*)** will calculate the number raised to the power specified.

**=POWER(2,3)** will return **8** (that is,  $2^3$ )

(The same result could be achieved by using **=2^3**)

Now suppose that you want to calculate  $\sqrt{3^2 + 4^2}$ . You could store 3 in cell **A1**, and 4 in cell **B1**. The required formula would then be

**=SQRT(POWER(A1,2)+POWER(B1,2))**

Here **POWER** has been used as an argument in the **SQRT** function.

## 12 Errors

Occasionally you may get an error value in a cell. This happens when Excel cannot deal sensibly with the formula you have supplied. There are several different errors caused in different ways as shown in the table below.

Error	Explanation
-------	-------------

#DIV/0!	You have tried to divide a number by zero (probably because the cell referred to as divisor is a blank cell).
#N/A	No information is available for the calculation you wish to do. This may be deliberate — see below.
#NAME?	A text string has not been enclosed in double quotes, <i>or</i> , you entered a name in a formula that is not in the Define Name dialog box list (see section 14.3).
#NULL!	You included a space between two ranges in a formula to indicate an intersection, but the ranges have no common cells.
#NUM!	You supplied an invalid argument to a worksheet function, <i>or</i> , the result of a formula is too large or too small to be represented in the worksheet.
#REF!	You deleted a range of cells whose references are included in a formula.
#VALUE!	You entered a mathematical formula that refers to a text entry.

If you apply the Text format to a cell that contains a formula, that formula will be thought of as text and displayed as such in the cell. If there is a formula in another cell which refers to this text-formatted cell, the result in that cell will be either the text value itself (for a formula that is just a direct reference and does not involve calculations) or the error value #VALUE!.

**Tip:** The procedure described above can be a useful way of seeing the effect of removing a formula without actually deleting it.

- Apply the text format to the cell containing the formula.
- Click the formula bar and press **Enter** (to recalculate the worksheet and display the text value).
- Look for any resulting #VALUE! errors.

To restore the formula

- Apply a numeric format to the cell.
- Click the formula bar and press **Enter**.

**Note:** You can use the **Edit | Find** command to locate error values.

## 12.1 Trapping errors using logical functions

If a formula refers to a cell containing an error value, that formula will also return an error value. So, it can be useful to determine whether a cell contains an error value.

There are three functions that enable you to trap errors:

**General forms:**

```
=ISERR(value)
=ISERROR(value)
=ISNA(value)
```

Here *value* is usually a reference to a cell or range but it can be a number, a formula, or literal text. If it is a range, only the cell in the same row or column is tested.

ISERR tests for all errors except #N/A

ISERROR tests for all errors including #N/A

ISNA tests for #N/A values only

All three functions return either TRUE or FALSE. These functions are useful as conditional tests in IF functions.

**Example:**

**=IF(ISERROR(C2/C3),0,C2/C3)**

Here, if there is going to be a problem trying to divide the contents of C2 by the contents of C3, the ISERROR function will return TRUE so the value 0 will be returned by the IF function. If ISERROR returns FALSE, the calculation C2/C3 will be carried out and that answer returned.

## 12.2 Tracing errors

If you have a worksheet that is displaying errors, it is possible to trace each one back to its source. If you would like to experiment with this,

- 1 Open a new workbook (or insert a new blank worksheet in **Data\_for\_Calculations**).
- 2 Enter 3 in **B8** and 0 in **B9** (that is a zero not a letter O).
- 3 In **C11** enter

**=B8/B9**

This will produce a #DIV/0! error.

- 4 In **E8** enter

**=C11+567**

That will give the same error.


- 5 In **F10** enter

**=2\*E8**

Again, you will see an error value.

- 6 From the **T**ools menu select **Formula Auditing**, and then **S**how **Formula Auditing Toolbar**.
- 7 Click in **F10** (the latest cell containing an error).
- 8 Click the **Trace Error** button on the **Formula Auditing** toolbar (seventh button from the left).

The resulting display is shown below. Red arrows trace the error chain and blue arrows show the cells used in the first of the formulae to give an error.



- 9 Remove the arrows by clicking on the **Remove All Arrows** button (sixth button from the left).
- 10 Close the **Formula Auditing** toolbar.

### 13 Using labels instead of cell references

If you are careful, you can sometimes use labels instead of cell references in a formula. Such formulae are called **natural language** formulae. It is probably best to look at a specific example.

- 1 Click on the **NamedRefs** tab of the **Data\_for\_Calculations** workbook.

The labels at the top of the columns (**Area1**, **Area2**, **Area3** and **Area4**) refer to the cells below them (**D6:D11**, **E6:E11**, **F6:F11** and **G6:G11**).

The labels to the left of the rows (**January**, **February**, and so on) refer to the cells to the right of them (**D6:G6**, **D7:G7**, and so on).

- 2 Click in **D13** and type

**=sum(Area1)**

- 3 Press **Enter**.

You should get the answer 306.

- 4 Copy the formula across the range **E13:G13**.

When you enter text into a formula, Excel looks for column and row labels that match that text. It then decides what to do.

If the label mentioned in the formula is in the same column (or row) as the cell containing the formula, Excel uses the full range of contiguous cells that are next to the label — *below* a column label; *to the right* of a row label.

If the label is not in the same row or column as the cell containing the formula, Excel uses the single cell at the intersection of the labelled column (or row) and the row (or column) of the cell containing the formula.

For example, the intersection of **Area4** and **June** is cell **G11**.

So, another way of writing the formula **=G11** is **=Area4 June** with a space between **Area4** and **June** (to represent the intersection operator).

- For practice, fill in the rest of the table — **Average sales for each region, Total sales for each month** and **Average sales for each month** — using labels.

You should get the following values:

	A	B	C	D	E	F	G	H	I	J	K
1											
2											
3		blank column	<b>Number of houses sold</b>					blank column			
4	blank row										
5			Region	Area1	Area2	Area3	Area4		Total sales for each month	Average sales for each month	
6			Month								
7			January	34	45	29	24		132	33	
8			February	36	41	23	20		120	30	
9			March	47	59	31	27		164	41	
10			April	51	61	29	231		372	93	
11			May	60	80	51	29		220	55	
12	blank row		June	78	92	59	35		264	66	
13			Total sales for each region	306	378	222	366				
14			Average sales for each region	51	63	37	61				
15											
16											

You will have noticed that the table containing the original data is surrounded by blank rows and columns. When using natural language formulae, leave blank rows and columns between the table and any cells containing calculations involving natural language references.

If you would like to see the effect of entering a value in one of the blank regions:

- Note the value in **D13** (306).
- Click in **D12**, type

**2004**

and press **Enter**.

The values in **D13** and **D14** are not automatically updated. To force a recalculation in **D13**:

- Double-click in **D13** and press **Enter**.

The result in **D13** will change to **2310** because the 2004 is now included in the calculation.

- Double-click in **D14** and press **Enter**.

That result will change to **577.5** because the 2004 and the 2310 are now included in the calculation.

- Use the **Undo** button to get back to the stage where **D12** was blank.

It is also possible to define labels and have Excel automatically substitute them for cell references.

## 14 Working with names

You have already seen how to build a formula using

- cell references
- column and row labels (natural language formulae — see section 13)

Another approach is to give names to cells and ranges of cells, and use those names in the formulae. A name that has been defined in one worksheet is available for use in all the other worksheets of that particular workbook.

### 14.1 Names of cells and ranges

There are some rules about naming cells and ranges:

- all names must begin with a letter or underscore character; the remaining characters can be letters, numbers, periods and underscore characters
- spaces are not allowed; use an underscore or period to separate words
- a name can contain up to 255 characters (but can not be selected from the Name box if it is longer than 253 characters)
- names can contain upper and lower case letters but are not case sensitive (Sales is equivalent to SALES)
- names that could be misinterpreted as cell references can not be used

### 14.2 Defining a name using the Name box

Remember that a single cell is selected by clicking in it; a range is selected by highlighting it.

Give the name **Cake** to cell **D19** as follows:

- 1 Make sure that the worksheet **NamedRefs** is active.
- 2 Select cell **D19** by clicking in it.

Note that the cell reference **D19** is displayed in the **Name** box at the left-hand end of the formula bar.

- 3 Click in that **Name** box.
- 4 Type

**Cake**

and press **Enter**.

Note how the word **Cake** is now showing in the **Name** box.

If a range has been named, its name will only appear in the **Name** box when the entire range is selected.

Next,

- 1 Give the name **Tea** to cell **E19**.

- 2 Click in **F19** and type

**=Cake+Tea**

This will give the same result as entering the formula **=D19+E19** (£1.20).

**Important:** When you define a name for a cell, the name of the worksheet is part of that definition and the cell reference is **absolute**. So, the actual definition of the name for **D19** is **NamedRefs!\$D\$19**.

### 14.3 Creating names from text cells

If there are labels in the row and/or column adjacent to a range of cells that you wish to name, they can be used as names.

Suppose you want to name the range **D25:D28**.

- 1 Select cells **D24:D28** (be sure to include the **Number of tins** heading).
- 2 From the **I**nsert menu, select **N**ame and then **C**reate.

In the **Create Names** dialog box, the **T**op row box will already be selected (has a tick).

- 3 Click **OK**.
- 4 Click on the down-arrow in the **N**ame box at the left-hand end of the formula bar and check that this new name is included.

To use this named range (**Number\_of\_tins**),

- 1 Click in **C31**.
- 2 Type  
**=sum(Number\_of\_tins)**  
and press **Enter**.

The result should be 215.

So that you can use it later,

- 1 Give the name **Cost\_of\_tin** to the range **E25:E28**.

**Note:** If you need to give a name to cells on more than one worksheet by using a 3D reference, use **I**nsert | **N**ame | **D**efine — see the online Help.

### 14.4 Pasting names into formulae

Although you could now use these defined names by typing them in to your formulae, if the names are long, that approach can be tedious and prone to errors. It is often easier to paste the names into formulae.

To fill in the **Total value** column (with the calculation *number of tins multiplied by the cost of one tin*):

- 1 Click in **F25**.

- 2 Type  
=
- 3 From the **I**nsert menu, select **N**ame and then **P**aste.
- 4 In the **Paste Name** dialog box, select **Number\_of\_tins** and click **OK**.
- 5 Type  
\*
- 6 Press the **F3** key (a quick way to get to the **Paste Name** dialog box).
- 7 Select **Cost\_of\_tin** and click **OK**.
- 8 Your formula is now complete so press the **Enter** key (£125.16).
- 9 Copy that formula down through cells **F26:F28**.
- 10 Click in **F27** and check that the formula is correct.

#### 14.5 Applying names to existing formulae

The formula in **C34** was entered without using names.

- 1 Click in **C34** and note that the formula is

**=AVERAGE(E25:E28)**

To make this formula use existing names:

- 2 From the **I**nsert menu, select **N**ame and then **A**pply.
- 3 Select **Cost\_of\_tin** (make sure that only that one is highlighted).
- 4 Click **OK**.
- 5 Note how the formula now reads

**=AVERAGE(Cost\_of\_tin)**

**Note:** If you need to select more than one name in that **Apply Names** dialog box, just hold down the **Shift** key as you select the names.

#### 14.6 Moving to a named cell or range

You can quickly move to a cell or range that has been named by any of the following methods:

- Click the down-arrow next to the **Name** box and select the name from the drop-down list.
- Type the name in the **Name** box.
- Select **Go To** from the **E**dit menu, click on the name and then on **OK**.

## 14.7 Producing a list of names

If you become very enthusiastic about naming cells, you may find it useful to produce a list of names and the cells to which they refer. As an example,

- 1 Click in **C38**.
- 2 Select **I**nsert | **N**ame | **P**aste.
- 3 In the **Paste Name** dialog box, click on the **Paste L**ist button.

## 15 Working with arrays

An array is an organised rectangular list of values. It can consist of:

- one row, several columns
- several rows, one column
- several rows, several columns

An **array formula** takes **array arguments** and can produce either single or multiple results.

A rectangular block of cells sharing the same array formula is called an **array range**.

When entering an array formula, you

- Select the array range where the results are to be stored.
- Type in your array formula.
- Press the **Ctrl**, **Shift** and **Enter** keys together (and then release them).

Note that you *do not* just press **Enter**.

Try working through some examples.

### 15.1 Examples

- 1 Click on the **Arrays** tab of the workbook.

The red and blue numbers can be thought of as arrays.

#### **Multiply all the x numbers in B3:B8 by 2**

Obviously you could enter the formula **=B3\*2** in **C3** and then copy that formula down through **C4:C8**. Try using arrays instead:

- 1 Highlight the range of cells **C3:C8**.
- 2 Type the following formula (but don't press the Enter key)

**=B3:B8\*2**

- 3 Press the **Ctrl**, **Shift** and **Enter** keys.
- 4 Have a look at the formula on the formula bar.

You will see that Excel has added a pair of curly brackets to indicate that it is an array formula.

### Add all the x numbers in B3:B8 to the y numbers in F3:F8

1 Highlight the range **D3:D8**.

2 Type

**=B3:B8+F3:F8**

3 Press the **Ctrl, Shift** and **Enter** keys together.

4 Check that the answers are correct.

### Multiply the numbers in B12:C13 by those in F12:G13

1 Highlight the range **D16:E17**.

2 Type

**=B12:C13\*F12:G13**

3 Press the **Ctrl, Shift** and **Enter** keys together.

Check that you understand what Excel has done. The first number in the green array was multiplied by the first number in the orange array and the result stored in the first cell of the new array, and so on.

That is clearly very different from matrix multiplication.

### Matrix multiplication

1 Highlight the range **D20:E21**.

2 Type

**=mmult(B12:C13,F12:G13)**

3 Press the **Ctrl, Shift** and **Enter** keys together.

4 Note the different results.

	<b>Array multiplication</b>		2	9	
			8	4	
	<b>Matrix multiplication</b>		14	6	
			20	10	

## 15.2 Changing cells in an array range

You cannot edit, clear or move individual cells in an array range. They have to be treated as a single entity. Also, cells cannot be inserted or deleted.

You can, however, format individual cells in an array and copy and paste them elsewhere.

To **select** an array,

- click in any cell belonging to that array and press the **Ctrl** and **/** keys together (click in **C5** and try that)

or

- select *all* the cells in the array.

To **edit** an array, carry out the following steps:

- 1 Select the array.
- 2 Activate the formula bar.
- 3 Edit the formula.
- 4 Press **Ctrl**, **Shift** and **Enter**.

### 15.3 Array constants

It is possible that at some time you may wish to work with a constant array defined by you. This could be made up of numbers, text or logical values. The array must be enclosed by curly brackets **{ }**. Commas are used to separate the values in different columns; semicolons are used to separate the values in different rows.

For example, the array

1	2	3
5	7	9

can be represented by **{1,2,3;5,7,9}**

So, if you wanted to multiply some values in the range **T100:V101** by that constant array, you could use **=T100:V101\*{1,2,3;5,7,9}**

## 16 Miscellaneous

### 16.1 Recalculation

#### Manual recalculation

If you have a large workbook involving lots of formulae, and you change the value in a cell, Excel's automatic recalculations can take a while to complete. To save time, you may like to consider changing to manual recalculation. Then Excel will not recalculate until you tell it to.

- 1 Select **T**ools | **O**ptions and click on the **C**alculation tab.
- 2 In the **C**alculation section, click beside **M**anual.
- 3 Note the position of the **C**alc **N**ow (**F9**) button that recalculates all your open workbooks and the **C**alc **S**heet button that recalculates just the active worksheet.
- 4 Click **OK**.

When you want Excel to recalculate, press

- **F9** — to recalculate all the cells in all the worksheets that are affected by the changes made since the last recalculation.
- **Shift** and **F9** together — to recalculate only the active worksheet

or use the **Calc Now (F9)** and **Calc Sheet** buttons mentioned above.

Even with manual recalculation, Excel will probably recalculate your entire workbook when you save it. If you want to prevent this,

- 1 Select **O**ptions from the **T**ools menu.
- 2 Click on the **C**alculation tab.
- 3 De-select the **R**ecalculate before save option.

If you select the **Automatic except tables** option, Excel will automatically recalculate everything except data tables (referred to in section 18 about What-if analysis).

### Workbooks from earlier versions of Excel

When you open a workbook that was created in the current version, Excel recalculates just those formulae that depend on cells that have changed.

However, when you open a workbook created in an earlier version of Excel, the workbook is completely recalculated. *All* the formulae are recalculated regardless of whether they depend on cells that have changed or not. This can take a while, but after the workbook has been saved in the current version it will be quicker to open. Should you wish, you can at any time force a full recalculation of a workbook by clicking the **Calculate Full** button (add it to your toolbar) or by pressing **Ctrl+Alt+F9**.

## 16.2 Formula bar formatting

When you are entering a long complicated formula, it can be made easier to read by inserting spaces and line breaks in appropriate places. This does not affect the calculation. Press the **Alt** and **Enter** keys together to insert a line break.

## 16.3 Replacing formulae with their resulting values

Sometimes you may want to keep a particular result of a formula rather than allow it to change when the values on which the formula depends are changed.

If you have a cell that contains a formula and you would like the cell to contain the resulting value (rather than the formula):

- 1 Click in the cell.
- 2 Select **C**opy from the **E**dit menu.
- 3 Select **P**aste **S**pecial from the **E**dit menu.
- 4 In the resulting **Paste Special** dialog box, select **V**alues.
- 5 Click on **OK**.

6 Press the **Enter** key.

If you want to change just part of the formula to a value:

- 1 Highlight the cell reference in the formula on the formula bar.
- 2 Press **F9**.
- 3 Press **Enter** to lock in the value.

Click **Undo** if you want to restore the formula to its original state.

#### 16.4 Circular references

If you enter a formula that depends on its own value, this is called a circular reference and Excel will display an error message. Just occasionally, a circular reference is intentional and Excel can be made to iterate a specified number of times. See the Help for more details.

#### 16.5 3-D formulae

A formula in a worksheet can refer to:

- other cells in that worksheet
- cells in a different worksheet within the same workbook
- cells in a different workbook

A **3-D reference** consists of a cell (or range) reference preceded by a range of worksheet names.

Suppose you have a workbook containing the following four *adjacent* worksheets:

**Sheet2** holds data for England, with a total value in **H2**

**Sheet3** holds data for Scotland, with a total value in **H2**

**Sheet4** holds data for Wales, with a total value in **H2**

**Sheet5** holds data for Ireland, with a total value in **H2**

In addition,

**Sheet1** contains a summary of the information in those sheets.

The grand total of those four **H2** values could then be entered in **Sheet1** with

**=SUM(Sheet2:Sheet5!H2)**

The functions that can be used with 3-D references are

<b>SUM</b>	<b>MIN</b>	<b>MAX</b>	<b>AVERAGE</b>
<b>PRODUCT</b>	<b>COUNTA</b>	<b>VAR</b>	<b>STDEV</b>
<b>STDEVP</b>	<b>COUNT</b>	<b>VARP</b>	

#### 16.6 Euro currency tools

Excel has a **Euro Currency Tools** add-in program that provides

- a standard toolbar button € for formatting euro values
- a button and a menu command for converting euros to/from other euro member currencies
- a EuroValue toolbar for viewing converted values
- a EUROCONVERT worksheet function for converting currencies

This add-in has been enabled on the Networked PC service.

If you are using a stand-alone PC,

- 1 Select **Add-Ins** from the **Tools** menu.
- 2 Select the **Euro Currency Tools** check box.

See the built-in help for more details.

**Note:** For Excel to recognise euro values, your operating system must support the euro sign. Windows 2000 Professional, Windows NT 4.0 with Service Pack 4 or later, Windows Millenium Edition and Windows 98 all have this support.

## 16.7 R1C1 reference style

In this document, cells references have contained a column letter and a row number. However, there is another way of referring to cells, known as the R1C1 reference style, where row *and* column numbers are used.

For example, **R20C5** means row **20**, column **5**, in other words, cell **E20**.

In this notation, a relative cell reference is described in terms of the position of that cell relative to the cell containing the formula.

For example,

**=RC[-2]+RC[-1]**

means *take* the contents of the cell in the same row but two columns to the left *and add* the contents of the cell in the same row but one column to the left.

A relative reference must include square brackets around the number in the reference.

	<b>[+n]</b>	<b>[-n]</b>
<b>R</b>	<b>n</b> rows below the formula cell	<b>n</b> rows above the formula cell
<b>C</b>	<b>n</b> columns to the right of the formula cell	<b>n</b> columns to the left of the formula cell

**R1C1** references were not used in this document but there may be occasions when you will find this approach useful (perhaps when writing macros). If you are interested in trying **R1C1** references,

- 1 Select **Options** from the **Tools** menu.
- 2 Click the **General** tab.

- 3 Select the **R1C1** reference style and click **OK**.
- 4 Have a look at some of the formulae you have entered.
- 5 Turn the option off when you have finished experimenting.

## 17 Analysis Toolpack

Excel provides an Analysis Toolpack add-in — a set of data analysis tools for use in statistical and engineering analyses.

### 17.1 Analysis ToolPak not installed

If you decide that you would like to use these tools but the **Data Analysis** command is not on the **Tools** menu, you need to install the Analysis ToolPak in Microsoft Excel as follows:

- 1 From the **Tools** menu, select **Add-Ins**.

If the **Analysis ToolPak** is not listed in the **Add-Ins** dialog box, click **Browse** and locate the drive, folder name, and file name for the **Analysis ToolPak** add-in, **Analys32.xll** — usually located in the **Library\Analysis** folder — or run the **Setup** program if it isn't installed.

- 2 Select **Analysis ToolPak**.
- 3 Click **OK**.

**Note:** Add-ins selected in the **Add-Ins** dialog box remain active until you remove them.

### 17.2 Using the Analysis ToolPak

You can use the **ToolPak** functions just as you would any other Excel function.

- 1 Select **Data Analysis** from the **Tools** menu.
- 2 Select a tool in the **Analysis Tools** box and click **OK**.
- 3 Supply information and select options in the dialog box that opens.

If a formula that uses a **ToolPak** function returns a **#NAME?** error value,

- 1 From the **Tools** menu, select **Add-Ins**.
- 2 Select **Analysis ToolPak-VBA** from the **Add-Ins available** list.
- 3 Click **OK**.

If the Analysis ToolPak-VBA is not listed in the Add-Ins available list you can install it by double-clicking the Microsoft Excel Setup icon.

## 18 What-if analysis

One of the most useful benefits when using a spreadsheet is the ability to carry out what-if analyses. When Excel is configured to recalculate

automatically, you can change the value in a cell and immediately see the effect of this change on any cells that depend on the value you have changed.

Two different kinds of data tables, sometimes called sensitivity tables, can be created by selecting **T**able from the **D**ata menu. The first table is based on *one variable* and finds out how changing that value affects several formulae that use it. The other is based on *two variables* and tests the effect on just one formula.

When making decisions you will probably find that *more than two variables* are involved. Providing you have no more than 32 variables you can use the **S**cenario **M**anager — select **S**cenarios from the **T**ools menu. A scenario is a particular set of possible values assigned to your variable cells. The Scenario Manager is a powerful tool that can be used to compare, contrast, combine and look at various possibilities.

If you have a problem that has a known target value, and that value depends on only one unknown variable, try using the **G**oal **S**eek command. This is available from the **T**ools menu.

If your problem involves more than one variable cell, the **S**olver is available to find combinations of those variables that will minimise or maximise the target cell. The Solver is an add-in. When installed, it is available from the **T**ools menu.

## 19 Other Excel documentation

Other ITS documentation about Excel includes:

*Guide 33: An introduction to Microsoft Excel 2003*

*Guide 34: Creating charts in Microsoft Excel 2003*

*Guide 36: Lists and data management in Microsoft Excel 2003*

*Guide 39: Introduction to using macros in Microsoft Excel 2003*